
Alloy is a language and a toolkit for ex-
ploring the kinds of structures that arise 
in many software designs. This brief arti-
cle aims to give a flavor of Alloy in action, 
to summarize how Alloy has been used to 
date, and thereby to give you a sense of 
how you might use it in your own soft-
ware design work.

Formal Design Languages
Software involves structures of many 
sorts: architectures, database schemas, 
network topologies, ontologies, and so 
on. When you design a software system, 
you need to be able to express the struc-
tures that are essential to the design, and 
to check that they have the properties you 
expect.

You can express a structure by sketch-
ing it on a napkin. That’s a good start, but 
it’s limited. Informal representations give 
inconsistent interpretations, and they 
can’t be analyzed mechanically. So people 
have turned to formal notations that de-
fine structure and behavior precisely and 
objectively, and that can exploit the pow-
er of computation.

By using formality early in develop-
ment, you can minimize the costs of am-
biguity and get feedback on your work by 
running analyses. The most popular ap-
proach that advocates this is agile devel-
opment, in which the formal representa-
tion is code in a traditional programming 
language and the analysis is conventional 
unit testing.

As a language for exploring designs, 
however, code is imperfect. It’s verbose 
and often indirect, and it doesn’t allow 
partial descriptions in which some details 
are left to be resolved later. And testing, 
as a way to analyze designs, leaves much 

to be desired. It’s notoriously incomplete, 
and burdensome, since you need to write 
test cases explicitly. And it’s very hard to 
use code to articulate design without get-
ting mired in low level details (such as the 
choice of data representations).

An alternative, which has been ex-
plored since the 1970s, is to use a design 
language built not on conventional ma-
chine instructions but on logic. Partiality 
comes for free because, rather than listing 
each step of a computation, you write a 
logical constraint saying what’s true after, 
and that constraint can say as little or as 
much as you please. To analyze such a 
language, you use specialized algorithms 
such as model checkers or satisfiability 
solvers (more on these below). This usu-
ally requires much less effort than test-
ing, since you only need to express the 
property you want to check rather than a 
large collection of cases. And the analy-
sis is much more complete than testing, 
because it effectively covers all (or almost 
all) test cases that you could have written 
by hand.

What Came Before: Theorem Provers 
and Model Checkers

To understand Alloy, it helps to know a 
bit about the context in which it was de-
veloped, and the tools that existed at the 
time.

Theorem provers are mechanical aids 
for constructing mathematical proofs. 
To apply a theorem prover to a software 
design problem, you formulate some in-
tended property of the design, and then 
attempt to prove the theorem that the 
property follows from the design. The-
orem provers tend to provide very rich 
logics, so they can usually express any 
property you might care about, at least 
about states and state transitions—more 
dynamic properties can require a tempo-
ral logic that theorem provers don’t typi-

cally support directly. Also, because they 
generate mathematical proofs, which can 
be checked by tools that are smaller and 
simpler than the tool that finds the proof, 
you can be confident that the analysis is 
sound.

On the other hand, the combination of 
an expressive logic and sound proof has 
meant that finding proofs cannot gener-
ally be automated. So theorem provers 
usually require considerable effort and 
expertise from the user, often orders of 
magnitude greater than the effort of con-
structing a formal design in the first place. 
Moreover, failure to find a proof does not 
mean that a proof does not exist, and the-
orem provers don’t provide counterexam-
ples that explain concretely why a theo-
rem is not valid. So theorem provers are 
not so useful when the intended property 
does not hold—which unfortunately is 
the common case in design work.

Model checkers revolutionized design 
analysis by providing exactly the fea-
tures theorem provers lacked. They offer 
push-button automation, requiring the 
user to give only the design and prop-
erty to be checked. They allow dynam-
ic properties to be expressed (through 
temporal logics), and generate counter-
examples when properties do not hold. 
Model checkers work by exploring the 
space of possible states of a system, and 
if that space is large, they may require 
considerable computational resources 
(or may fail to terminate). The so-called 
“state explosion” problem arises because 
model checkers are often used to analyze 
designs involving components that run in 
parallel, resulting in an overall state space 
that grows exponentially with the number 
of components.

Alloy was inspired by the successes and 
limitations of model checkers. For designs 
involving parallelism and simple state 
(comprising boolean variables, bound-

ed integers, enumerations and fixed-size 
arrays), model checkers were ideal. They 
could easily find subtle synchronization 
bugs that appeared only in rare scenari-
os that involved long traces with multiple 
context switches, and therefore eluded 
testing.

For hardware designs, model checkers 
were often a good match. But for software 
designs they were less ideal. Although 
some software design problems involve 
this kind of synchronization, often the 
complexity arises from the structure 
of the state itself. Early model checkers 
(such as SMV [9]) had limited expressive-
ness in this regard, and did not support 
rich structures such as trees, lists, tables 
and graphs.

Explicit state model checkers, such 
as SPIN [14], and later Java Pathfinder 
[37], allowed designs with rich state to 
be modeled, but, despite providing sup-
port for temporal properties, gave little 
help for expressing structural ones. To 
express reachability (for example that two 
social media users are connected by some 
path of friend edges), you would typical-
ly need to code an explicit search, which 
would have to be executed at every point 
at which the property was needed. Also, 
explicit state model checkers have limit-
ed support for partiality (since the model 
checker would have to conduct a costly 
search through possible next states to 
find one satisfying the constraints).

Particularly hard for all model checkers 
are the kinds of designs that involve a con-
figuration of elements in a graph or tree 
structure. Many network protocols are 
designed to work irrespective of the ini-
tial configuration (or of the configuration 
as it evolves), and exposing a flaw often 
involves not only finding a behavior that 
breaks a property but also finding a con-
figuration in which to execute it.

Even the few model checkers that can 
express rich structures are generally not 
up to this task. Enumerating possible 
configurations is not feasible, because 
the number of configurations grows su-
per-exponentially: if there are n nodes, 
there are 2n×n ways to connect them.

Alloy’s Innovations
Alloy brought a new kind of design lan-
guage and analysis, made possible by 
three innovations.

Relational logic. Alloy uses the same 
logic for describing designs and proper-
ties. This logic combines the for-all and 
exists-some quantifiers of first-order log-
ic with the operators of set theory and re-
lational calculus.

The idea of modeling software designs 
with sets and relations had been pio-
neered in the Z language [32]. Alloy in-
corporated much of the power of Z, while 
simplifying the logic to make it more 
tractable.

First, Alloy allows only first-order struc-
tures, ruling out sets of sets and relations 
over sets, for example. This changes how 
designs are modeled, but not what can be 
modeled; after all, relational databases 
have flourished despite being first order.

Second, taking advantage of this re-
striction, Alloy’s operators are defined in 
a very general way, so that most expres-
sions can be written with just a few oper-
ators. The key operator is relational join, 
which in conventional mathematics only 
applies to binary relations, but in Alloy 
works on relations of any arity. By using 
a dot to represent the join operator, Al-
loy lets you write dereferencing expres-
sions as you would in an object oriented 
programming language, but gives these 
expressions a simple mathematical inter-
pretation. So, as in Java, given an employ-
ee e, a relation dept that maps employees 
to departments, and a relation manager 
that maps departments to their manag-
ers, e.dept.manager would give the man-
ager of e’s department. But unlike in Java, 
the expression will also work if e is a set of 
employees, or dept can map an employee 
to multiple departments, giving the ex-
pected result—the set of managers of the 
set of departments that the employees e 
belong to. The expression dept.manager is 
well defined too, and means the relation 
that maps employees to their managers. 
You can also navigate backwards, writing 
manager.m for the department(s) that m 
manages.

(A note for readers interested in lan-
guage design: this flexibility is achieved 
by treating all values as relations—a set 
being a relation with one column, and a 
scalar being a set with one element—and 
defining a join operator that applies uni-
formly over a pair of relations, irrespec-
tive of their arity. In contrast, other lan-
guages tend to have multiple operators, 

implicit coercions or overloading to ac-
commodate variants that Alloy unifies.)

Alloy was influenced also by modeling 
languages such as UML. Like the class 
diagrams of UML, Alloy makes it easy to 
describe a universe of objects as a classi-
fication tree, with each relation defined 
over nodes in this tree. Alloy’s dot oper-
ator was inspired in part by the naviga-
tional expressions of OCL (the Object 
Constraint Language [39] of UML), but 
by defining the dot as relational join, Al-
loy dramatically simplifies the semantics 
of navigation.

Small scope analysis. Even plain first-or-
der logic (without relational operators) is 
not decidable. This means that no algo-
rithm can exist that could analyze a soft-
ware design written in a language like Al-
loy completely. So something has to give. 
You could make the language decidable, 
but that would cripple its expressive pow-
er and make it unable to express even the 
most basic properties of structures (al-
though exciting progress has been made 
recently in applying decidable fragments 
of first-order logic to certain problems 
[29]). You could give up on automation, 
and require help from the user, but this 
eliminates most of the benefit of an analy-
sis tool; analysis is no longer a reward for 
constructing a design model, but a major 
extra investment beyond modeling.

The other option is to somehow limit 
the analysis. Prior to Alloy, two approach-
es were popular. Abstraction reduces the 
analysis to a finite number of cases, by in-
troducing abstract values that each corre-
spond to an entire set of real values. This 
often results in false positives that are 
hard to interpret, and in practice picking 
the right abstraction calls for considerable 
ingenuity. Simulation picks a finite num-
ber of cases, usually by random sampling, 
but it covers such a small part of the state 
space that subtle flaws elude detection.

Alloy offered a new approach: running 
all small tests. The designer specifies a 
scope that bounds each of the types in the 
specification. A scope of 5, for example, 
would include tests involving at most 5 
elements of each type: 5 network nodes, 5 
packets, 5 identifiers, and so on.

The rationale for this is the small scope 
hypothesis, which asserts that most bugs 
can be demonstrated with small counter-
examples. That means that if you test for 
all small counterexamples, you are likely 



to find any bug. Many Alloy case studies 
have confirmed the hypothesis, by per-
forming an analysis in a variety of scopes 
and showing, retrospectively, that a small 
scope would have sufficed to find all the 
bugs that were discovered.

Translation to SAT. Even with small 
scopes, the state space of an Alloy model 
is fiendishly large. The state comprises a 
collection of variables whose values are 
relations. Just one binary relation in a 
scope of 5 has 5 × 5 = 25 possible edges, 
and thus 225 possible values. A very small 
design might have 5 such relations, giving 
(225)5 possible states—about 1037 states. 
Even checking a billion cases per second, 
such an analysis would take many times 
the age of the universe.

Alloy therefore does not perform an 
explicit search, but instead translates the 
design problem to a satisfiability prob-
lem whose variables are not relations but 
simple bits. By flipping bits individually, a 
satisfiability (SAT) solver can usually find 
a solution (if there is one) or show that 
none exists by examining only a tiny por-
tion of the space.

Alloy’s analysis tool is essentially a com-
piler to SAT, which allows it to exploit the 
latest advances in SAT solvers. The suc-
cess of SAT solvers has been a remarkable 
story in computer science: theoreticians 
had shown that SAT was inherently in-
tractable, but it turned out that most of 
the cases that arise in practice can be 
solved efficiently. So SAT went from be-
ing the archetypal insoluble problem used 

to demonstrate the infeasibility of oth-
er problems to being a soluble problem 
that other problems could be translated 
to. Alloy also applies a variety of tactics 
to reduce the problem prior to solving, 
most notably adding symmetry breaking 
constraints that save the SAT solver from 
considering cases that are equivalent to 
one another.

Example: Modeling Origins
To see Alloy in action, let’s explore the de-
sign of an origin-tracking mechanism for 
web browsers. The model shown here is a 
toy version of a real model that exposed 
several serious flaws in browser security 
[1]. Although it cuts corners and is unre-
alistic in some respects, it does capture 
the spirit and style of the original model, 
and is fairly representative of how Alloy is 
often used.

First, some background for those unfa-
miliar with browser security. Cross-site 
request forgery (CSRF) is a pernicious 
and subtle attack in which a malicious 
script running in a page that the user has 
loaded makes a hidden and unwanted re-
quest to a website for which the user is al-
ready authenticated. This may happen ei-
ther because the user was enticed to load 
a page from a malicious server, or because 
a supposedly safe server was the subject 
of a cross-site scripting attack, and served 
a page containing a malicious script. Such 
a script can issue any request the user 
can issue; one of the first CSRF vulnera-
bilities to be discovered, for example, al-

lowed an attacker to change the delivery 
address for the user’s account in a DVD 
rental site. What makes CSRF particularly 
problematic is that the browser sends au-
thentication credentials stored as cookies 
spontaneously when a request is issued, 
whether that request is made explicitly by 
the user or programmatically by a script.

One way to counter CSRF is to track 
the origins of all responses received from 
servers. In our example, the browser 
would mark the malicious script as orig-
inating at the malicious or compromised 
server. The subsequent request made by 
that script to the rental site server—the 
target of the attack—would be labeled as 
having this other origin. The target server 
can be configured so that it only accepts 
requests that originate directly from the 
user (for example, by the user entering 
the URL for the request in the address 
bar), or from itself (for example, from a 

1 abstract sig EndPoint { }

2 sig Server extends EndPoint {
3  causes: set HTTPEvent
4  }

5 sig Client extends EndPoint { }

6 abstract sig HTTPEvent {
7  from, to, origin: EndPoint
8  }

9 sig Request extends HTTPEvent {
10  response: lone Response
11  }

12 sig Response extends HTTPEvent {
13  embeds: set Request
14  }

15 sig Redirect extends Response {
16  }

fig. 1  Structure declarations
fig. 2  Data model from declarations 

17 fact Directions {
18  Request.from + Response.to in Client
19  Request.to + Response.from in Server
20  }

21 fact RequestResponse {
22  all r: Response | one response.r
23  all r: Response |
24   r.to = response.r.from
25   and r.from = response.r.to
26  all r: Request |
27   r not in r.^(response.embeds)
28  }

29 fact Causality {
30  all e: HTTPEvent, s: Server |
31   e in s.causes iff e.from = s or
32    some r: Response |
33     e in r.embeds and r in s.causes
34  }

35 fact Origin {
36  all r: Response, e: r.embeds |
37   e.origin = r.origin
38  all r: Response | r.origin =
39   (r in Redirect implies
40    response.r.origin else r.from)
41  all r: Request |
42   no embeds.r implies
43    r.origin in r.from
44  }

45 pred EnforceOrigins (s: Server) {
46  all r: Request |
47   r.to = s implies
48    r.origin = r.to or r.origin = r.from
49  }

fig. 3  Fact and predicate declarations



script embedded in a page previously sent 
by the target server). As always the devil 
is in the details, and we shall see that a 
plausible design of this mechanism turns 
out to be flawed.

Here are some features to look out for 
in this model, which distinguish Alloy 
from many other approaches:
· A rich structure of objects, classifica-

tion and relationships;
· Constraints in a simple logic that ex-

ploits the relations and sets of the 
structure, avoiding the kind of low lev-
el structures (arrays and indices, etc.) 
that are often needed in model check-
ers;

· Capturing dynamic behavior without 
any need for a built-in notion of time 
or state;

· Intended properties to check ex-
pressed in the same language as the 
model itself;

· An abstract style of modeling that in-
cludes only those aspects essential to 
the problem at hand.

We start by declaring a collection of sig-
natures (Fig. 1). A signature introduces 
a set of objects and some fields that re-
late them to other objects. So Server, for 
example, will represent the set of server 
nodes, and has a field causes that asso-
ciates each server with the set of HTTP 
events that it causes.

Keywords (or their omission) indicate 
the multiplicity of the relations between 
objects: thus each HTTP event has ex-
actly one from endpoint, one to endpoint, 
and one origin endpoint (line 7); each re-
quest has at most one response (line 10, 
with lone being read as “less than or equal 
to one”); and each response embeds any 
number of requests (line 13).

Objects are, mathematically, just atom-
ic identifiers without any internal struc-
ture. So the causes relation includes tu-
ples of the form (s, e) where the value of 
s is some atomic identifier representing a 

server object, and the value of e is some 
atomic identifier representing an event.

Fields are declared in signatures to al-
low a kind of object-oriented mindset. Al-
loy supports this by resolving field names 
contextually (so that field names need 
not be globally unique), and by allowing 
“signature facts” (not used here) that are 
implicitly scoped over the elements of a 
signature and their fields. But don’t be 
misled into thinking that there is some 
kind of complex object semantics here. 
The signature structure is only a conve-
nience, and just introduces a set and some 
relations.

The extends keyword defines one sig-
nature as a subset of another. An abstract 
signature has no elements that do not be-
long to a child signature, and the exten-
sions of a signature are disjoint. So the 
declarations of EndPoint, Server and Client 
imply that the set of endpoints is parti-
tioned into servers and clients: no server 
is also a client, and there is no endpoint 
that is neither client nor server. A relation 
defined over a set applies over its subsets 
too, so the declaration of from, for exam-
ple, which says that every HTTP event is 

from a single endpoint, implies that the 
same is true for every request and re-
sponse. (Alloy is best viewed as untyped. 
It turns out that conventional program-
ming language types are far too restric-
tive for a modeling language. Alloy thus 
allows expressions such as HTTPEvent.re-
sponse, denoting the set of responses to 
any events, but its type checker rejects an 
expression such as Request.embeds which 
always denotes an empty set [12].)

The Alloy Analyzer can generate a 
graphical representation of the sets and 
relations from the signature declarations 
(Fig. 2); this is just an alternative view and 
involves no analysis.

Moving to the substance of what the 
model actually means:
· The from and to fields are just the 

source and destination of the event’s 
packet.

· For a response r, the expression r.em-
beds denotes a set of requests that 
are embedded as JavaScript in the re-
sponse; when that response is loaded 
into the browser, the requests are exe-
cuted spontaneously.

· A redirect is a special kind of response 
that indicates that a resource has 
moved, and spontaneously issues a re-
quest to a different server; this second 
request is modeled as an embedded re-
quest in the redirect response.

· The origin of an event is a notion com-
puted by the browser as a means of 
preventing cross-site attacks. As we’ll 
see later, the idea is that a server may 
choose to reject an event unless it orig-
inated at that server or at a browser.

· The cause of an event is not part of the 
actual state of the mechanism. It is in-
troduced in order to express the essen-
tial design property: that an evil server 
cannot cause a client to send a request 
to a good server.

Now let’s look at the constraints (Fig. 3). 
If there were no constraints, any behavior 
would be possible; adding constraints re-
stricts the behavior to include only those 
that are intended by design.

The constraints are grouped into sepa-
rate named facts to make the model more 
understandable:
· The Directions fact contains two con-

straints. The first says that every re-
quest is from, and every response is to, 
a client; the second says that every re-
quest is to, and every response is from 

50 check {
51 no good, bad: Server {
52  good.EnforceOrigins
53  no r: Request |
54    r.to = bad and r.origin in Client
55  some r: Request |
56    r.to = good and r in bad.causes
57  }
58 } for 5

fig. 4  Check command

fig. 5  Counterexample for check of Fig. 4 



a server. These kinds of constraints can 
be written in many ways. Here we’ve 
chosen to use expressions denoting 
sets of endpoints—Request.from for 
the set of endpoints that requests are 
from, eg. But we could equally well 
have written a constraint like

from in 
 Request -> Client + Response -> Server

to say that the from relation maps re-
quests to clients and responses to 
servers. Or in a more familiar but less 
succinct style, we could have used 
quantifiers:

all r: Request | r.from in Client 
all r: Response | r.from in Server

(which constrains only the range of 
the relations, which is sufficient in this 
case since the declarations constrain 
their domains).

· The RequestResponse fact defines some 
basic properties of how requests and 
responses work: that every response is 
from exactly one server (line 22); that 
every response is to the endpoint its 
request was from, and from the end-
point its request was to (line 23); and 
that a request cannot be embedded in 
a response to itself (line 26). Two ex-
pressions in these constraints merit 
explanation. The expression response.r 
exploits the flexibility of the join oper-
ator to navigate backwards from the 
response r to the request it responds 
to; it could equivalently be written 
r.~response using the transpose op-
erator ~. The expression r.^(response.
embeds) starts with the request r, and 
then applies to it one or more naviga-

tions (using the closure operator ^) of 
following the response and embeds re-
lations, as if we’d written instead the 
infinite expression

r.response.embeds 
+ r.response.embeds.response.embeds 
+ r.response.embeds.response.embeds 
  .response.embeds 
+ …

defining the requests embedded in the 
response to r, the requests embedded 
in the response to the requests em-
bedded in the response to r, and so on. 
(Equivalently, r.^p is the set of nodes 
reachable from r in the graph whose 
edges correspond to the relation p.)

· The Causality fact defines the causes 
relation. It says that an event is caused 
by a server if and only if it is from that 
server, or is embedded in a response 
that the server causes.

· The Origin fact describes the ori-
gin-tracking mechanism. Each con-
straint defines the origin of a different 
kind of HTTP event. The first (line 36) 
says that every embedded request e has 
the same origin as the response r that 
it is embedded in. The second (line 38) 
defines the origin of a response: it says 
that if the response is a redirect, it has 
the same origin as the original request, 
and otherwise its origin is the server 
that the response came from. The third 
(line 41) handles a request that is not 
embedded: its origin is the endpoint it 
comes from (which will usually be the 
browser).

Finally, EnforceOrigins is a predicate that 
can be applied to a server, indicating that 
it chooses to enforce the origin header, 
allowing incoming requests only if they 
originate at that server, or at the client 
that sent the request.

With all this in place—the structure of 
endpoints and messages, the rules about 
how origins are computed and used, and 
the definition of causality—we can define 
a design property to check (Fig. 4).

The keyword check introduces a com-
mand that can be executed. This com-
mand instructs the Alloy Analyzer to 
search for a refutation for the given con-
straint. In this case, the constraint asserts 
the non-existence of a cross-site request 
forgery attack; refuting this will show that 
the origin mechanism is not designed 
correctly, and an attack is possible.

The constraint says that there are no 
two servers, good and bad, such that the 
good server enforces the origin header 
(line 52), there are no requests sent di-
rectly to the bad server that originate 
in the client (line 53), and yet there is 
some request to the good server that was 
caused by the bad server (line 55).

Analysis Results: Finding Bugs
The Alloy Analyzer finds a counterexam-
ple (Fig. 5) almost instantaneously—in 
30ms on my 2012 Mac Book (with a 2.6 
GHz i7 processor and 16GB of RAM).

The counterexample can be displayed 
in various ways—as text, as a table, or as 
a graph whose appearance can be cus-
tomized. I’ve chosen the graph option, 
and have selected which objects are to 
appear as nodes (just the events and the 
servers), which relations are to appear as 
edges (those between events, and causes), 
and I’ve picked colors for the sets and re-
lations. I’ve also chosen to use the Skolem 
constants (witnesses that the analyzer 
finds for the quantified variables) good 
and bad to label the servers.

Reading the graph from the top, looking 
just at the large rectangles representing 
the HTTP events, we see that a request 
(Req1) was sent from a client to the good 
server. The response (Resp) embeds a re-
quest (Req0) that is sent to the bad server; 
this is a cross-site request which won’t be 
rejected because the bad server accepts 
incoming requests irrespective of origin. 
The bad server’s response is to send a re-
direct whose embedded request (Req2) is 
received by the good server. (Note that 
the numbering of objects is arbitrary: 
Req1 actually happens before Req0.)

Now looking at the server nodes and 
the events they cause, we see that, as ex-

fig. 7  A simulated instance 

fig. 6  A bogus counterexample 



pected, the good server caused the re-
sponse to the first request, and the bad 
server caused the redirect and its subse-
quent embedded request. The problem is 
the mismatch between cause and origin 
in the last request (Req2): we can see that 
it was caused by the bad server, but it was 
labelled as originating at the good server. 
In other words, the origin tracking design 
is allowing a cross-site request forgery by 
incorrectly identifying the origin of the 
request in the redirect.

The solution to this problem turns out 
to be non-trivial. Updating the origin 
header after each redirect would fail for 
websites that offer open redirection; a 
better solution is to list a chain of end-
points in the origin header [1].

Agile Modeling
As I mentioned earlier, our model is rep-
resentative of many Alloy models. But the 
way I presented it was potentially mis-
leading. In practice, users of Alloy don’t 
construct a model in its entirety and then 
check its properties. Instead, they pro-
ceed in a more agile way, growing the 
model and simulating and checking it as 
they go.

Take, for example, the constraint on 
line 26 of Fig. 3. Initially, I hadn’t actually 
noticed the need for this constraint. But 
when I ran the check for the first time 
(without this constraint), the analyzer 
presented me with counterexamples such 
as the one shown in Fig. 6, in which the 
response to a request is the very response 
in which the request is embedded!

One way to build a model, exploiting 
Alloy’s ability to express and analyze very 
partial models, is to add one constraint at 
a time, exploring its effect. You don’t need 
to have a property to check; you can just 
ask for an instance of the model satisfying 
all the constraints.

Doing this even before any explicit 
constraints have been included is very 
helpful. You can run just the data model 
by itself and see a series of instances that 
satisfy the constraints implicit in the dec-
larations. Often doing this alone exposes 
some interesting issues. In this case, the 
first few instances include examples with 
no HTTP events, and with requests and 
responses that are disconnected.

To get more representative instances, 
you can specify an additional constraint 
to be satisfied. For example, the command

run {some response}

will show instances in which the response 
relation has some tuples. The first one 
generated (Fig. 7) shows a request with a 
response that is a redirect from the same 
source as the request, and sent to an end-
point that is also its origin, and it includes 
an orphaned redirect unrelated to any re-
quest! These anomalies immediately sug-
gest enrichments of the model.

When we developed Alloy, we under-
estimated the value of this kind of sim-
ulation. As we experimented with Alloy, 
however, we came to realize how helpful it 
is to have a tool that can generate provoc-
ative examples. These examples invari-
ably expose basic misunderstandings, not 
only about what’s being modeled but also 
about which properties matter. It’s essen-
tial that Alloy provides this simulation 
for free: in particular, you don’t need to 
formulate anything like a test case, which 
would defeat the whole point.

Growing a model in a declarative lan-
guage like Alloy is very different from 
growing a program in a conventional 
programming language. A program starts 
with no behaviors at all, and as you add 
code, new behaviors become possible. 
With Alloy, it’s the opposite. The empty 
model, since it lacks any constraints, al-
lows every possible behavior; as you add 
constraints, behaviors are eliminated.

This allows a powerful style of incre-
mental development in which you only 
add constraints that are absolutely essen-
tial for the task at hand—whether that’s 
eliminating pathological cases or ensur-
ing that a design property holds.

Typically a model includes both a de-
scription of the mechanism being de-
signed and some assumptions about 
the environment in which it operates. 
Our model does not separate these rig-
orously, but where brevity is not such a 
pressing concern, it would be wise to do 
so. We could separate, for example, the 
constraints that model the setting and 
checking of the origin field from those 
that describe what kinds of requests and 
responses are possible.

Obviously, the less you assume about 
the environment, the better, since every 
assumption you make is a risk (since it 
may turn out to be untrue). In our mod-
el, for example, we don’t require every 
request to have a response. It would be 

easy to do—just change the declaration 
of response in line 10 of Fig. 1 by drop-
ping the lone keyword—but would only 
make the result of the analysis less gen-
eral. Likewise, the less you constrain the 
mechanism, the better. Allowing multiple 
behaviors gives implementation freedom, 
which is especially important in a distrib-
uted setting.

Simulation matters for a more pro-
found reason. Verification—that is, 
checking properties—is often overrated 
in its ability to prevent failure. As Chris-
topher Alexander explains [2], designed 
artifacts usually fail to meet their purpos-
es not because specifications are violated 
but because specifications are unknown. 
The “unknown unknowns” of a software 
design are invariably discovered when the 
design is finally deployed, but can often 
be exposed earlier by simulation, espe-
cially in the hands of an imaginative de-
signer.

Verification, in contrast, is too narrow-
ly focused to produce such discoveries. 
This is not to say that property checking is 
not useful—it’s especially valuable when a 
property can be assured with high confi-
dence using a tool such as Alloy or a mod-
el checker or theorem prover (rather than 
by testing). But its value is always contin-
gent on the sufficiency of the property it-
self, and techniques that help you explore 
properties have an important role to play.

Uses of Alloy
Hundreds of papers have reported on 
applications of Alloy in a wide variety of 
settings. Here are some examples to give a 
flavor of how Alloy has been used.

Critical systems. A team at the Univer-
sity of Washington constructed a depend-
ability case [18] for a neutron radiothera-
py installation. They devised an ingenious 
technique for verifying properties of code 
against specifications using lightweight, 
pluggable checkers. The end-to-end de-
pendability case was assembled in Alloy 
from the code specifications, proper-
ties of the equipment and environment, 
and the expected properties, and then 
checked using the Alloy Analyzer. The 
analysis found several safety-critical flaws 
in the latest version of the control soft-
ware, which the researchers were able to 
correct prior to its deployment. For a full 
description, see a recent research report 



[30] and additional information on the 
project’s website [36].

Network protocols. Pamela Zave, a re-
searcher at AT&T, has been using Alloy 
for many years to construct and analyze 
models of networking, and for designing 
a new unifying network architecture. In 
a major case study, she analyzed Chord, 
a distributed hash table for peer-to-peer 
applications. The original paper on Chord 
[33]—one of the most widely cited papers 
in computer science—notes that an inno-
vation of Chord was its relative simplicity, 
and consequently the confidence users 
can have in its correctness. By modeling 
and analyzing the protocol in Alloy, Zave 
showed that the Chord protocol was not, 
however, correct, and she was able to de-
velop a fixed version that maintains its 
simplicity and elegance while guarantee-
ing correct behavior [43]. Zave also used 
the explicit model checker SPIN [14] in 
this work, and wrote an insightful article 
explaining the relative merits of the two 
tools, and how she used them in tandem 
[42].

Web security. The demonstration exam-
ple of this paper is drawn from a real study 
performed by a research group at Berke-
ley and Stanford [1]. They constructed a 
library of Alloy models to capture various 
aspects of web security mechanisms, and 
then analyzed five different mechanisms, 
including: WebAuth, a web-based au-
thentication protocol based on Kerberos 
deployed at several universities including 
Stanford; HTML5 forms; the Cross-Ori-
gin Resource Sharing protocol; and pro-
posed designs for using the referer header 
and the origin header to foil cross-site 
attacks (of which the last is the basis for 
the example here). The base library was 
written in 2,000 lines of Alloy; the various 
mechanisms required between 20 and 
214 extra lines; and every bug was found 
within two minutes and a scope of 8. Two 
previously known vulnerabilities were 
confirmed by the analysis, and three new 
ones discovered.

Memory models. John Wickerson and 
his colleagues have shown that four 
common tasks in the design of memory 
models—generating conformance tests, 
comparing two memory models, check-
ing compiler optimizations, and checking 
compiler mappings—can all be framed 
as constraint satisfaction problems in 
Alloy [41]. They were able to reproduce 

automatically several results for C11 (the 
memory model introduced in 2011 for C 
and C++) and common compiler optimi-
zations associated with it, for the memory 
models of the IBM Power and Intel x86 
chips, and for compiler mappings from 
OpenCL to AMD-style GPUs. They then 
used their technique to develop and check 
a new memory model for Nvidia GPUs.

Code verification. Alloy can also be used 
to verify code, by translating the body of 
a function into Alloy, and asking Alloy to 
find a behavior of the function that vio-
lates its specification. Greg Dennis built a 
tool called Forge that wraps Alloy so that 
it can be applied directly to Java code an-
notated with JML specifications. In a case 
study application [10], he checked a vari-
ety of implementations of the Java collec-
tions list interface, and found bugs in one 
(a GNU Trove implementation). Dennis 
also applied his tool to KOA, an electron-
ic voting system used in the Netherlands 
that was annotated with JML specifica-
tions and had previously been analyzed 
with a theorem proving tool, and found 
several functions that did not satisfy their 
specifications [11].

Civil engineering. In one of the more 
innovative applications of Alloy, John 
Baugh and his colleagues have been ap-
plying Alloy to problems in large-scale 
physical simulation. They designed an 
extension to ADCIRC—an ocean circula-
tion model widely used by the U.S. Army 
Corps of Engineers and others for simu-
lating hurricane storm surge—that intro-
duces a notion of subdomains to allow 
more localized computation of changes 
(and thus reduced overall computational 
effort). Their extension, which has been 
incorporated into the official ADCIRC 
release, was modeled and verified in Al-
loy [7].

Alloy as a backend. Because Alloy of-
fers a small and expressive logic, along 
with a powerful analyzer, it has been ex-
ploited as a backend in many different 
tools. Developers have often used Alloy’s 
own engine, Kodkod [34], directly, rath-
er than the API of Alloy itself, because it 
offers a simpler programmatic interface 
with the ability to set bounds on rela-
tions, improving performance. Jasmine 
Blanchette’s Nitpick tool [8], for example, 
uses Kodkod to find counterexamples in 
Isabelle/HOL, saving the user the trouble 
of trying to prove a theorem that is not 

true, and the Margrave tool [26] analyzes 
firewall configurations. Last year, a team 
from Princeton and Nvidia built a tool 
that uses Alloy to synthesize security at-
tacks that exploit the Spectre and Melt-
down vulnerabilities [35].

Teaching. Alloy has been widely taught 
in undergraduate and graduate courses 
for many years. At the University of Min-
ho in Portugal, Alcino Cunha teaches an 
annual course on formal methods using 
Alloy, and has developed a web interface 
to present students with Alloy exercises 
(which are then automatically checked). 
At Brown University, Tim Nelson teach-
es Logic for Systems, which uses Alloy for 
modeling and analysis of system designs, 
and has become one of the most popular 
undergraduate classes. Because the Al-
loy language is very close to a pure rela-
tional logic, it has also been popular in 
the teaching of discrete mathematics, for 
example in a course that Charles Wallace 
teaches at Michigan Technological Uni-
versity [38] and appearing as a chapter in 
a popular textbook [15].

Alloy Extensions
Many extensions to Alloy—both to the 
language and to the tool—have been cre-
ated. These offer a variety of improve-
ments in expressiveness, performance 
and usability. For the most part, these 
extensions have been mutually incom-
patible, but a new open source effort is 
now working to consolidate them. There 
are too many efforts to include here, so 
we focus on representatives of the main 
classes.

Higher-order solving. The Alloy Analyz-
er’s constraint solving mechanism cannot 
handle formulas with universal quantifi-
cations over relations—that is, problems 
that reduce to “find some relation P such 
that for every relation Q…” This is exactly 
the form that many synthesis problems 
take, in which the relation P represents a 
structure to be synthesized, such as the 
abstract syntax tree of a program, and the 
relation Q represents the state space over 
which certain behaviors are to be verified. 
Alloy* [24] is an extension of Alloy that 
can solve such formulas, by generalizing 
a tactic known as counterexample-guided 
inductive synthesis that has been widely 
used in synthesis engines.

Temporal logic. Alloy has no built-in 
notion of time or dynamic behavior. On 



the one hand, this is an asset, because it 
keeps the language simple, and allows 
it to be used very flexibly. We exploited 
this in the example model of this paper, 
where the flow of time is captured in the 
response relation that maps each request 
to its response. By adding a signature for 
state, Alloy supports the specification 
style common in languages such as B, 
VDM and Z; and by adding a signature 
for events, Alloy allows analysis over trac-
es that can be visualized as series of snap-
shots. On the other hand, it would often 
be preferable to have dynamic features 
built into the language. Electrum [20] ex-
tends Alloy with a keyword var to indicate 
that a signature or field has a time-varying 
value, and with the quantifiers of linear 
temporal logic (which fit elegantly with 
Alloy’s traditional quantifiers). DynAlloy 
[31] offers similar functionality, but using 
dynamic logic instead, and is the basis of 
an impressive code analysis tool called 
TACO [13] that outperforms Forge (men-
tioned above) by employing domain-spe-
cific optimizations. No extension of Alloy, 
however, has yet addressed the problem 
of combining Alloy’s capacity for struc-
tural analysis with the ability of tradition-
al model checkers to explore long traces, 
so Alloy analyses are still typically limited 
to short traces.

Instance generation. The result of an Al-
loy analysis is not one but an entire set of 
solutions to a constraint-solving problem, 
each of which represents either a positive 
example of a scenario, or a negative ex-
ample, showing how the design fails to 
meet some property. The order in which 
these appear is somewhat arbitrary, being 
determined both by how the problem is 
encoded and the tactics of the backend 
SAT solver. Since SAT solvers tend to 
try false before true values, the instanc-
es generated tend to be small ones—with 
few nodes and edges. This is often desir-
able, but is not always ideal. Various ex-
tensions to the Alloy Analyzer provide 
more control over the order in which in-
stances appear. Aluminum [28] presents 
only minimal scenarios, in which every 
relation tuple is needed to satisfy the con-
straints, and lets the user add new tuples, 
automatically compensating with a (min-
imal) set of additional tuples required for 
consistency. Amalgam [27] lets users ask 
about the provenance of an instance, in-
dicating which subformula is responsible 

for requiring (or forbidding) a particular 
tuple in the instance. Another extension 
[21] of the Alloy Analyzer generates min-
imal and maximal instances, and choos-
ing a next instance that is as close to, or 
as far away from, the current instance as 
possible.

Better numerics. Alloy handles numeri-
cal operations by treating numbers as bit 
strings. This has the advantage of fitting 
into the SAT solving paradigm smoothly, 
and it allows a good repertoire of integer 
operations. But the analysis scales poorly, 
making Alloy unsuitable for heavily nu-
meric applications. The finite scopes of 
Alloy can also be an issue when a design-
er would like numbers to be unbounded. 
A possible solution is to replace the SAT 
backend with an SMT backend instead. 
This is challenging because SMT solvers 
have not traditionally supported relation-
al operators. Nevertheless, a team at the 
University of Iowa has recently extended 
CVC4, a leading SMT solver, with a theo-
ry of finite relations, and has promisingly 
demonstrated its application to some Al-
loy problems [23].

Configurations. Many Alloy models 
contain two loosely coupled parts, one 
defining a configuration (say of a network) 
and the other the behavior (say of sending 
packets). By iterating through configura-
tions and analyzing each independently, 
one can often dramatically reduce anal-
ysis time [22]. In some applications, a 
configuration is already fully or partially 
known, and the goal is to complete the 
instance—in which case searching for the 
configuration is a wasted effort. Kodkod, 
Alloy’s engine, allows the explicit defini-
tion of a “partial instance” to support this, 
but in Alloy itself, this notion is not well 
supported (and relies on a heuristic for 
extracting partial instances from formu-
las in a certain form). Researchers have 
therefore proposed a language extension 
[25] to allow partial instances to be de-
fined directly in Alloy itself.

How to Try Alloy
The Alloy Analyzer [3] is a free download 
available for Mac, Windows and Linux. 
The Alloy book [16] provides a gentle in-
troduction to relational logic and to the 
Alloy language, gives many examples of 
Alloy models, and includes a reference 
manual and a comparison to other lan-
guages (both of which are available on the 

book’s website [17]). The Alloy communi-
ty answers questions tagged with the key-
word alloy on StackOverflow, and hosts a 
discussion forum [5]. A variety of tutori-
als for learning Alloy are available online 
too, as well as blog posts with illustrative 
case studies and examples (eg [40, 19]). 
The model used in this paper is available 
(along with its visualization theme) in the 
Alloy community’s model repository [4].

